

## Chapter 2

# Expressions, Functions and Equations

This chapter discusses the difference between expressions, functions and equations, and how they are defined, manipulated and plotted in Maple. It also describes how to solve one or more equations. Briefly:

- An **expression** is simply a formula for computing something. It may involve zero, one or more variables, arithmetic operations and functions, as listed in Section 1.1. For example, (See Section 2.1.) the area of a circle of radius  $r$  is given by  $\pi r^2$  and entered as the Maple expression:

```
> Area:=Pi*r^2;
```

- A **function** is a rule for computing an output for each given input. It is usually specified *explicitly* by saying that the output (the dependent variable) is given by a some expression involving the input (the independent variable). For example, (See Section 2.1.) the area  $A$  of a circle is a function of its radius  $r$  given by the rule  $A(r) = \pi r^2$  and entered as the Maple (arrow-defined) function:

```
> A:=r->Pi*r^2;
```

However, a function may also be specified *implicitly* as the solution of some equation. For example, (See Section 4.4.) the cube root function is defined as the solution (for  $y$ ) of the equation  $x = y^3$ .

- Finally, an **equation** is a statement that two expressions are equal to each other. For example, (See Section 2.3.) the statement that the area of a circle is  $40 \text{ cm}^2$  is given by the equation  $\pi r^2 = 40$  and entered into Maple as:

```
> eq:=Pi*r^2=40;
```

NOTE: The Maple statement `Area:=Pi*r^2;` is *not* an equation because there is an assignment symbol `:=` but no equal sign.

The distinction between expressions, functions and equations may seem pedantic, but mathematics makes this distinction and Maple handles them very differently.

When you complete this chapter, please preview Chapter 11. This may help you avoid some of the typical problems with Maple.

## 2.1 Expressions and Functions: Definition and Use

**Definition and Evaluation of Expressions and Functions:** In the previous chapter, we saw that we can assign labels to numbers (such as `Area:=25*Pi`). In a similar way, labels can be assigned to expressions, functions and equations, (all of which may contain variables which may or may not have been previously assigned values). This is useful when repeatedly referring to a complicated formula. For example, the area of a circle is the expression

```
> Area := Pi*r^2;
```

$$Area := \pi r^2$$

You can find the area for a particular value of  $r$  (say  $r = 5$ ) by substituting the value into the area using either the `subs` command or the `eval` command:

```
> subs(r=5,Area); eval(Area,r=5);
```

$$25\pi$$

$$25\pi$$

The difference is that `eval` simplifies the answer while `subs` does not:

```
> subs(x=Pi/4,sin(x)); eval(sin(x),x=Pi/4);
```

$$\sin\left(\frac{\pi}{4}\right)$$

$$\frac{\sqrt{2}}{2}$$

The area of a circle can also be thought of as the function

```
> A := r -> Pi*r^2;
```

$$A := r \rightarrow \pi r^2$$

In this “arrow notation” for a function,  $r$  is the independent variable and  $\pi r^2$  is the expression which gives the rule by which to compute the value of the dependent variable  $A$ . (Notice that the arrow is typed as a minus and a greater than sign (->) with *no space*!) For a given value of  $r$ , the corresponding value of  $A$  is denoted by  $A(r)$ . So if  $r = 5$ , then  $A(5) = 25\pi$ . In Maple we compute

```
> A(5);
```

$$25\pi$$

Notice that

```
> A(r);
```

$$\pi r^2$$

is just the *expression* for the area.

CAUTION: Above, we said you substitute numbers into expressions by using the `subs` command. There is another way to do it (which is not as good). You can simply assign a value to the variable, e.g.

```
> r:=7;
                                r := 7
```

Then this value is automatically substituted into any quantity that uses this variable, e.g.

```
> Area;
                                49 π
```

The problem comes when you try to define another quantity which uses that variable. For example, the circumference of the circle is:

```
> Circum:=2*Pi*r;
                                Circum := 14 π
```

The value is automatically substituted here too. If you want the formula with a variable `r`, then you must first unassign `r` by entering:

```
> r:='r';
                                r := r
```

(Those are forward single quotes (').) Now `r` is back to a free variable and `Area` is back to its formula in terms of `r`:

```
> Area;
                                π r2
```

However, quantities which were entered while `r` had a value, retain that value, e.g.

```
> Circum;
                                14 π
```

This is probably not what you wanted. This means you need to define (or redefine) the quantity while `r` is a free variable:

```
> Circum:=2*Pi*r;
                                Circum := 2 π r
```

The moral is that it is best never to assign values to the free variables in your formulas; use `subs` or `eval` instead.

Expressions, functions and equations can involve more than one variable. For example, the volume of a cylinder of radius  $r$  and height  $h$  is the expression

```
> Volume:=Pi*r^2*h;
                                Volume := π r2 h
```

or the function

```
> V:=(r,h) -> Pi*r^2*h;
                                V := (r, h) → π r2 h
```

In this case, the volume involves two variables  $r$  and  $h$ .

Multiple substitutions can be entered at the same time. If simultaneous substitution is desired, then curly braces { } or square brackets [ ] must be used. For example, to substitute  $r = 2$  and  $h = 5$  into the expression for the volume of a cylinder, enter

```
> subs({r=2,h=5},Volume);
20 π
```

If you don't include the braces, the `subs` command makes the substitutions one at a time from left to right. For example, in the following

```
> subs(r=h,h=5,Volume);
125 π
```

`r` is first replaced by `h`, giving  $\pi h^3$ , and then `h` is replaced by 5 giving  $125\pi$ . However, in the following `r` is replaced by `h` and `h` is simultaneously replaced by 5:

```
> subs({r=h,h=5},Volume);
5 π h2
```

More realistic examples of simultaneous and sequential substitutions occur in Example 1 of Section 3.3 and Example 2 of Section 3.4.

If you use `eval` then only simultaneous substitution is possible:

```
> eval(Volume,{r=h,h=5});
5 π h2
```

Thus `subs` gives you more control than `eval`.

Functions of two or more variables are evaluated using the usual function notation:

```
> V(2,5);
20 π
```

Again notice that the *expression* for the volume can be recovered by applying the function to its variables:

```
> V(r,h);
π r2 h
```

**Using Expressions and Functions in Commands:** Some commands, like `expand` and `factor`, can only be applied to expressions. Others, like composition, denoted by `@`, can only be applied to functions. Still others, like `plot`, can be applied to either but the syntax is different. For example, the expression

```
> g:=2*x^3-5*x^2+x+2;
g := 2 x3 - 5 x2 + x + 2
```

can be factored by executing

```
> factor(g);
(x - 1) (x - 2) (2 x + 1)
```

However, the arrow-defined function

```
> f:= x -> 2*x^3-5*x^2+x+2;
```

$$f := x \rightarrow 2x^3 - 5x^2 + x + 2$$

can only be factored after you evaluate at  $x$  to produce the expression  $f(x)$ :

```
> factor(f(x));
```

$$(x - 1)(x - 2)(2x + 1)$$

On the other hand, you can compose the function  $f$  and the function `sin`, by typing `f@sin`. Its value is then

```
> (f@sin)(x);
```

$$2\sin(x)^3 - 5\sin(x)^2 + \sin(x) + 2$$

which is the same as

```
> f(sin(x));
```

$$2\sin(x)^3 - 5\sin(x)^2 + \sin(x) + 2$$

However, to compose the expression  $g$  and the expression `sin(x)` you must use the `subs` or `eval` command

```
> subs(x=sin(x),g);
```

$$2\sin(x)^3 - 5\sin(x)^2 + \sin(x) + 2$$

The `@` notation is much closer to standard math notation.

In the next section, we will discuss the `plot` command which has a different syntax for expressions and functions.

It should also be noted that the `subs` command is purely a symbolic replacement. Thus

```
> subs(x^2=2, x^4+x^2+x);
```

$$x^4 + 2 + x$$

simply replaces all occurrences of  $x^2$  by 2. To substitute for all  $x$ 's you must use

```
> subs(x=sqrt(2), x^4+x^2+x);
```

$$6 + \sqrt{2}$$

In addition, the `subs` command does *no* simplification. It must be followed by `simplify(%)`, if necessary. For example

```
> f:=(y^2-x)/x; subs(y=x,f); simplify(%)
```

$$f := \frac{y^2 - x}{x}$$

$$\frac{x^2 - x}{x}$$

$$x - 1$$

**Converting Between Expressions and Functions:** Since some commands only work with either a function or an expression, it is important to be able to change back and forth between them.

To convert from a function to an expression is easy. As mentioned above, if  $f$  is a function, then  $f(x)$  is the expression given by *applying*  $f$  to the variable  $x$ .

Conversely, to convert from an expression to a function, there are two options: The first method is simply to retype the formula (or copy and paste it) and insert the arrow notation  $x \rightarrow$  as we did in going from  $g$  to  $f$  above. The problem with this method is that the formula for  $f$  may not appear in the worksheet. It may be a complicated result of a long computation.

The second method involves the `unapply` command. When you *apply* a function to its variable, you get an expression. Conversely, when you *unapply* an expression to its variable, you get a function. For example, since  $g$  is the expression

```
> g;
      2 x3 - 5 x2 + x + 2
```

then

```
> h:=unapply(g,x);
      h := x → 2 x3 - 5 x2 + x + 2
```

is the corresponding function. The second entry in `unapply(g,x)` tells Maple that the variable in the function will be  $x$ . (There may be other variables in the expression besides  $x$ .) This method can be very helpful since you don't need to write out the expression for  $h$ .

CAUTION: The following assignment does not work!

```
> h:= x -> g;
      h := x → g
```

because then

```
> h(x);
      2 x3 - 5 x2 + x + 2
```

but

```
> h(2);
      2 x3 - 5 x2 + x + 2
```

also. You must use `unapply`!

If the function is to have more than two variables, then the additional variables appear as additional arguments to `unapply`. For example, the volume of a cylinder can be defined as the expression

```
> Volume:=Pi*r^2*h;
      Volume := π r2 h
```

and converted to a function using:

```
> V:=unapply(Volume,r,h);
      V := (r, h) → π r2 h
```

## 2.2 Expressions and Functions: Plots

This section explains how to plot expressions and functions with one or two variables using the `plot` and `plot3d` commands.

NOTE: These commands cannot plot *equations*. For that use the `implicitplot` and `implicitplot3d` commands discussed in Section 2.4.


**Two-Dimensional Plots:** To plot the expression

```
> g:=2*x^3-5*x^2+x+2;
```

$$g := 2x^3 - 5x^2 + x + 2$$

over the interval  $-2 \leq x \leq 3$ , use the `plot` command to execute

```
> plot(g,x=-2..3);
```



To plot the function

```
> f:=x->2*x^3-5*x^2+x+2;
```

$$f := x \rightarrow 2x^3 - 5x^2 + x + 2$$

over the same interval, execute

```
> plot(f,-2..3);
```

which gives the same plot. Note the difference in syntax. Since `f` is a function, it already knows its independent variable, so we don't need to (and must not) specify it with the interval. However, since `f(x)` is an expression, it is also valid to plot `f` using

```
> plot(f(x),x=-2..3);
```

which again gives the same plot. On the other hand, a mix of syntax will result in error messages or empty plots.

```
> plot(f,x=-2..3);
```

```
Error, (in plot) expected a range but received x = -2 .. 3
```

```
> plot(f(x),-2..3);
```

```
Warning, unable to evaluate the function to numeric values in the
region; see the plotting command's help page to ensure the calling
sequence is correct
```

```
Plotting error, empty plot
```

```
> plot(g,-2..3);
```

Warning, unable to evaluate the function to numeric values in the region; see the plotting command's help page to ensure the calling sequence is correct

Plotting error, empty plot


To graph two or more expressions or functions in the same plot with the same interval, enclose them in curly braces { } or square brackets [ ]. For example, to graph  $g$  (from above) and

```
> h:=2*x;
```

$$h := 2x$$


execute

```
> plot({g,h},x=-2..3);
```



If you enclose them in square brackets, then options (like color, linestyle and thickness) can be specified separately for each expression:

```
> plot([g,h],x=-2..3, linestyle=[SOLID,DOT], thickness=[3,1]);
```



NOTE: The two expressions must have the same variable and range. On the other hand, functions may have different variables. For example, to plot  $f$  (from above) and


```
> k:=t-> 10*sin(t);
```

$$k := t \rightarrow 10 \sin(t)$$



execute

```
> plot([f,k],-2..3, linestyle=[SOLID,DOT], thickness=[3,1]);
```



NOTE: The variable is not specified with the range.

If for some reason, you want to plot two expressions (or functions) with different ranges, then you cannot use a single `plot` command. You must make two separate plots and then combine them using the `display` command from the `plots` package. For example, to plot  $x^2$  on  $-2 \leq x \leq 1$  and  $2 - x$  on  $0 \leq x \leq 2$ , first plot them separately and label the plots:

```
> p1:=plot(x^2,x=-2..1):
```

```
> p2:=plot(2-x,x=0..2):
```

Notice we ended these commands with colons (rather than semicolons) to suppress output. (The use of semicolons would display small uninformative outputs. Try it.) Next we load the `plots` package


```
> with(plots);
```

[*animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d, conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d, densityplot, display, dualaxisplot, fieldplot, fieldplot3d, gradplot, gradplot3d, graphplot3d, implicitplot, implicitplot3d, inequal, interactive, interactiveparams, intersectplot, listcontplot, listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot, logplot, matrixplot, multiple, odeplot, pareto, plotcompare, pointplot, pointplot3d, polarplot, polygonplot, polygonplot3d, polyhedra-supported, polyhedraplot, rootlocus, semilogplot, setcolors, setoptions, setoptions3d, spacecurve, sparsematrixplot, surfdata, textplot, textplot3d, tubeplot*]

NOTE: Maple responds with a list of special plot commands, one of which is `display`. (In the future, we avoid this list by using a colon instead of a semicolon. Ignore any warning.)

Now we can `display` the two plots.

```
> display([p1,p2]);
```




**Three-Dimensional Plots:** We have discussed expressions and functions of two or more variables, but the most we can plot is a function of two variables. To plot the expression

```
> g:=y+sin(2*x+y);
```

$$g := y + \sin(2x + y)$$

over the region  $-5 \leq x \leq 5$ ,  $-5 \leq y \leq 5$ , use the `plot3d` command to execute

```
> plot3d(g,x=-5..5,y=-5..5);
```



To plot the function

```
> f:=(x,y)->y+sin(2*x+y);
```

$$f := (x, y) \rightarrow y + \sin(2x + y)$$

over the same region, execute

```
> plot3d(f,-5..5,-5..5);
```

which gives the same plot. Note the difference in syntax. For a function, you don't specify the independent variables with the intervals. However, since  $f(x,y)$  is an expression, it is also valid to plot  $f$  using

```
> plot3d(f(x,y),x=-5..5,y=-5..5);
```


which again gives the same plot.

As with 2-dimensional plots, there are a large number of options to modify the display of 3-dimensional plots. A complete list may be seen by executing

```
> ?plot3d,options
```

In particular, the option `axes=normal` will add axes, `scaling=constrained` will equalize the scale on the three axes and `style=patchnogrid` will remove the grid lines.


```
> plot3d(f,-5..5,-5..5, axes=normal, scaling=constrained,
> style=patchnogrid);
```



You can also modify many of these options by right-clicking in the plot and changing the option.

Several options cannot be changed from the right-click menu. For example, `grid=[#, #]` will modify the number of points computed in each direction (with the default being  $25 \times 25$ ) and `orientation=[ $\theta$ ,  $\phi$ ]` will rotate the plot where  $\theta$  and  $\phi$  are spherical coordinates with  $\phi$  measured down from the north pole:

```
> plot3d(f,-5..5,-5..5, grid=[5,10],
> orientation=[45,30]);
```



You can also *dynamically rotate the plot by clicking in the plot and dragging the mouse*. And you can see your current orientation by clicking in the plot and looking at the left of the plot toolbar. If you right click in the plot and select MANIPULATOR, you can change the action of dragging the mouse from ROTATE to SCALE or PAN. Try it!

## 2.3 Equations in One Variable

In Section 2.1, we defined the area of a circle as the expression

```
> Area:=Pi*r^2;
```

$$Area := \pi r^2$$

So the equation that the area of the circle is 40 may be entered and labeled by

```
> Area_eq:=Area=40;
```

$$Area\_eq := \pi r^2 = 40$$

Notice that the equality is denoted by an equal sign (=), while the assignment to a label is a colon-equal (:=). You can read off the left and right hand sides of an equation by using the `lhs` and `rhs` commands:

```
> lhs(Area_eq); rhs(Area_eq);
```

$$\pi r^2$$

$$40$$

When substituting values into an equation, Maple handles the equation like an expression by using the `subs` or `eval` command. Thus if  $r = a + b$ , then the area equation becomes:

```
> eval(Area_eq, r=a+b);
```

$$\pi (a + b)^2 = 40$$

Maple has two commands for solving equations, `solve` and `fsolve`, providing exact and approximate solutions, respectively. To solve for the radius of the circle whose area is 40, execute

```
> r0:=solve(Area_eq,r);
```

$$r0 := \frac{2\sqrt{10}}{\sqrt{\pi}}, -\frac{2\sqrt{10}}{\sqrt{\pi}}$$

Maple does not know that  $r$  is supposed to be positive; so it has given two solutions. These may be referred to as `r0[1]` and `r0[2]`. Thus the correct solution is

```
> r1:=r0[1];
```

$$r1 := \frac{2\sqrt{10}}{\sqrt{\pi}}$$

To get a (floating point) decimal value, you can use `evalf`:

```
> evalf(r1);
```

$$3.568248232$$

Alternatively, you can get the (floating point) decimal solutions directly using

```
> r2:=fsolve(Area_eq,r);
```

$$r2 := -3.568248232, 3.568248232$$

and select the positive one using

```
> r3:=r2[2];
```

$$r3 := 3.568248232$$

Now consider a more substantial example:

EXAMPLE: Find the point(s) of intersection of the parabola  $y = x^2$  and the line  $y = 1 - 2x$ .


SOLUTION: First let's define and plot the two expressions to get an idea where they intersect:

```
> f:=x^2; g:=1-2*x;
```

$$f := x^2$$

$$g := 1 - 2x$$

```
> plot({f,g},x=-3..3);
```



There are two solutions. To find them exactly you need to solve the equation

```
> f=g;
```

$$x^2 = 1 - 2x$$

```
> x0:=solve(f=g,x);
```

$$x0 := -1 + \sqrt{2}, -1 - \sqrt{2}$$

Notice that we labeled the solutions `x0`. So the first and second solutions may be referred to as `x0[1]` and `x0[2]`. Alternatively, you can immediately give two names to the two solutions by using Maple's multiple assignment command:

```
> x1,x2:=solve(f=g,x);
```

$$x1, x2 := -1 + \sqrt{2}, -1 - \sqrt{2}$$

TIP: When the result of a Maple command is two or more expressions separated by commas (called an expression sequence), you may assign them to the same number of labels separated by commas, as above.

Returning to the example, the  $y$  coordinates are

```
> y1:=subs(x=x1,g); y2:=subs(x=x2,g);
```

$$y1 := 3 - 2\sqrt{2}$$

$$y2 := 3 + 2\sqrt{2}$$

In decimals, the two points are

```
> pt1:=evalf([x1,y1]); pt2:=evalf([x2,y2]);
```

$$pt1 := [0.414213562, 0.171572876]$$

$$pt2 := [-2.414213562, 5.828427124]$$

which agrees with the plot.

Exact solutions to complicated equations can be difficult or impossible to find. For example, enter the equation

```
> eq:=x^3-2*x^2+x-3=0;
```

$$eq := x^3 - 2x^2 + x - 3 = 0$$

Solving this equation with the command `solve(eq,x)`; will yield three very complicated solutions. (Try this. The letter  $I$  is used to denote  $\sqrt{-1}$ . So two solutions are actually complex numbers.)

To get decimal approximations to these solutions, change the coefficients of the equation from exact integers to floating-point decimals.

```
> eq:=x^3-2.0*x^2+x-3.0=0;
      eq := x3 - 2.0x2 + x - 3.0 = 0
```

Now solving the equation will yield nicer (but only approximate) decimal answers.

```
> solve(eq,x);
2.174559410, -0.08727970515 + 1.171312111 I, -0.08727970515 - 1.171312111 I
```

Note that the answers involving  $I = \sqrt{-1}$  are complex.

Some equations are impossible to solve exactly. In fact, there is a famous theorem in mathematics that states that there is no formula (analogous to the quadratic formula) for finding roots of polynomials of fifth degree or higher. Consider the following equation.

```
> eq:=x^7+3*x^4+2*x-1=0;
      eq := x7 + 3x4 + 2x - 1 = 0
```

Try finding the exact solutions using Maple.

```
> sol:=solve(eq,x);
sol := RootOf(%1, index = 1), RootOf(%1, index = 2), RootOf(%1, index = 3),
RootOf(%1, index = 4), RootOf(%1, index = 5), RootOf(%1, index = 6),
RootOf(%1, index = 7)
%1 := _Z7 + 3_Z4 + 2_Z - 1
```

Maple's response indicates that it does not know how to solve this equation exactly. Maple knows there are seven roots and uses the notation `RootOf(..., index= #)` to denote the seven roots. In this situation, use Maple's `evalf` or `fsolve` commands to find approximate solutions.

```
> evalf(sol);
0.4414177090, 0.7266822765 + 1.126611893 I, 0.2452048032 + 0.8980515885 I,
-1.192595934 + 0.1793080390 I, -1.192595934 - 0.1793080390 I,
0.2452048032 - 0.8980515885 I, 0.7266822765 - 1.126611893 I
```

```
> fsolve(eq,x);
0.4414177090
```

Notice that `solve` with `evalf` gives all the complex roots while `fsolve` only gives the real root, unless you add the `complex` parameter to the command:

```
> fsolve(eq,x,complex);
```

```
-1.192595934 - 0.1793080390 I, -1.192595934 + 0.1793080390 I,
0.2452048032 - 0.8980515885 I, 0.2452048032 + 0.8980515885 I,
0.4414177090, 0.7266822765 - 1.126611893 I, 0.7266822765 + 1.126611893 I
```

When the equation is not polynomial, it is best to use the `fsolve` command in conjunction with a plot. Consider the equation

```
> eq:=x^2+1/x-1/x^2=0;
```

$$eq := x^2 + \frac{1}{x} - \frac{1}{x^2} = 0$$


Solving this equation with `fsolve` will yield one solution.

```
> sol1:=fsolve(eq,x);
```

```
sol1 := -1.220744085
```

However, a plot of the expression  $f := x^2 + 1/x + 1/x^2$  indicates that there is another solution.

```
> plot(x^2+1/x-1/x^2,x=-2..2,y=-20..20);
```



This second solution can be found by modifying the `fsolve` command to include a range:

```
> sol2:=fsolve(eq,x=0..1);
```

```
sol2 := 0.7244919590
```

TIP: The `fsolve` command will find all the real roots of a polynomial, but as the previous example shows, more complicated equations should be handled by a plot together with specifying a range in the `fsolve` command. Alternatively, the `avoid` option can be used to avoid known solutions:

```
> sol2:=fsolve(eq,x,avoid={x=sol1});
```

```
sol2 := 0.7244919590
```

## 2.4 Equations in Two or More Variable


Equations can involve any number of variables. For example, the equation of a circle of radius 2, centered at the origin is

```
> circle:=x^2+y^2=4;
```

$$\text{circle} := x^2 + y^2 = 4$$


The simple `plot` command cannot plot equations. To plot an equation involving two variables, use the `implicitplot` command in the `plots` package. (Unlike the `plot` command, the `implicitplot` command does not have a default range. The  $x$ - and  $y$ -ranges must be entered each time.)

```
> with(plots):
> implicitplot(circle, x=-2..2, y=-2..2);
```



The graph of  $x^2 + y^2 = 4$  should be a circle. If you get an ellipse, as above, it is because the scale is not the same on both axes. To fix this, add the option `scaling=constrained`. You can also increase the resolution by adding the option `grid=[p,q]` where  $p$  specifies the number of grid points along the  $x$ -axis and  $q$  specifies the number of grid points along the  $y$ -axis. The default values are  $p=q=25$ . (Don't make  $p$  and  $q$  too large so the computer does not get overloaded.  $p=q=75$  is usually sufficient.)

```
> implicitplot(circle, x=-2..2, y=-2..2, grid=[49,49],
> scaling=constrained);
```




You can add the graph of a line to this plot by using curly braces `{ }`

```
> line:=2*y=x-2;
```

$$\text{line} := 2y = x - 2$$

```
> implicitplot({circle,line}, x=-2..2, y=-2..2,
> scaling=constrained);
```





Maple can find the points of intersection. To find them approximately, click in the plot on an intersection point. The coordinates will appear in a box at the top left of the Maple window, just under the FILE menu. To find them exactly use the `solve` command with curly braces `{ }`:

```
> sol:=solve({circle,line},{x,y});
```

$$sol := \{x = 2, y = 0\}, \{x = \frac{-6}{5}, y = \frac{-8}{5}\}$$

NOTE: In the `solve` command, Maple requires curly braces enclosing the equations `{circle,line}` and the variables `{x,y}`. The output `sol` consists of two solutions and each is also in the form of a set. Within each set, the variables may appear in a random order. You may refer to the two solutions as `sol[1]` and `sol[2]` but to get them in a useful form as points, execute the following:

```
> p1:=subs(sol[1],[x,y]); p2:=subs(sol[2],[x,y]);
```

$$p1 := [2, 0]$$

$$p2 := [\frac{-6}{5}, \frac{-8}{5}]$$

Maple can also solve systems with larger numbers of equations.

EXAMPLE: Solve the system of equations

$$4x - 2y + z = 8$$

$$x - 3y + 2z = 1$$

$$3x + y - 2z = 9$$

SOLUTION: Enter the equations:

```
> eq1:=4*x-2*y+z=8; eq2:=x-3*y+2*z=1; eq3:=3*x+y-2*z=9;
```

$$eq1 := 4x - 2y + z = 8$$

$$eq2 := x - 3y + 2z = 1$$

$$eq3 := 3x + y - 2z = 9$$

Solve the equations:

```
> sol:=solve({eq1,eq2,eq3},{x,y,z});
```


$$sol := \{x = 2, y = -1, z = -2\}$$

Convert the solution into a point:

```
> pt:=subs(sol, [x,y,z]);
      pt := [2, -1, -2]
```

You can also plot equations with three variables using the `implicitplot3d` command in the `plots` package. For example, here is the plot of a sphere.

```
> sphere:=x^2+y^2+z^2=9;
      sphere := x^2 + y^2 + z^2 = 9
> implicitplot3d(sphere, x=-3..3, y=-3..3, z=-3..3,
> scaling=constrained);
```



## 2.5 Summary

- Distinguish between Maple expressions, functions, and equations.
- Know how to form a Maple expression, how to use arrow notation to form a Maple function and how to enter an equation.
- Know the proper syntax to evaluate a Maple expression or equation (using `subs` or `eval`) or a Maple function (by applying it to a value).
- Convert a Maple function to an expression by applying the function to its variable. Convert a Maple expression to a function by using the `unapply` command.
- Distinguish between the assignment command which changes the values in memory and `subs` or `eval` which only affect what is on the screen. Distinguish between an assignment (`:=`) and an equation (`=`).
- Know the syntax for plotting a Maple expression using `plot` or `plot3d`, a Maple function using `plot` or `plot3d` and a Maple equation using `implicitplot` or `implicitplot3d`.
- Know how to use `solve` to find exact solutions to one or more equations and how to use `evalf` to convert exact solutions to (floating point) decimal approximations.
- When `solve` produces expressions involving `RootOf`'s, know how to use `evalf` to obtain decimal approximations.
- Know how to use `fsolve` to produce approximate (floating point) decimal solutions. Know that `fsolve` will find all real solutions for polynomial equations but only one real solution for non-polynomial equations.

- Know how to find additional decimal solutions by using a plot in conjunction with `fsolve` and a variable range or an `avoid` option.
- Be able to display two graphs simultaneously in the same plot either by enclosing several expressions or functions in curly braces or square brackets or by using the `display` command.
- Right-click in the plot window or use `plot` command options to adjust the display of plots.

## 2.6 Exercises

1. Enter an expression that describes the area of a square in terms of its diagonal length. Use the variable name `areasq` for the area of the square and let  $d$  be the diagonal length. Use the `subs` command to compute the area of the square when  $d = 3$ ,  $d = 3.2$  and  $d = 4\pi$ .
2. Enter an expression that describes the area of a circle in terms of its circumference. Use the variable name `areacir` for the area of the circle and let  $C$  be the circumference. Use the `eval` command to compute the area of the circle when  $C = 3$ ,  $C = 3.2$  and  $C = 4\pi$ . Find a decimal approximation using `evalf`.
3. Enter an expression that describes the volume of a cube in terms of its diagonal length. Use the variable name `vol` for the volume of the cube and let  $d$  be the diagonal length. (If you assigned a value to  $d$  in Exercise 1, be careful to unassign it before you enter this expression.) Use the `subs` command to compute the volume of the cube when  $d = 3$ ,  $d = 3.2$  and  $d = 4\pi$ .
4. Use `unapply` to convert the expressions from Exercises 1–3 into Maple functions. Evaluate these functions when the independent variable (either  $d$  or  $C$ ) is 3, 3.2 and  $4\pi$ .

TIP: Execute `restart` now and before most exercises.

5. The volume of a cone of radius  $r$  and height  $h$  is  $V = \frac{1}{3}\pi r^2 h$ . Write the volume as an arrow defined function of the two variables  $r$  and  $h$ . Evaluate this function at  $r = 4$  and  $h = 3$ . Convert the volume function into an expression in terms of radius  $R$  and height  $H$ .
6. Define an expression for the volume of a prism whose height is  $h$  and whose base is an equilateral triangle of side length  $s$ . Use `unapply` to convert this into a function of the two variables  $s$  and  $h$ . Compute the volume of the prism when  $s = 4$ ,  $h = 3$  and when  $s = 3.1$ ,  $h = 3.9$ . Note the area of an equilateral triangle of side  $s$  is  $\frac{\sqrt{3}s^2}{4}$ .

7. A large tank initially contains 200 gallons of salt water with a concentration of 1/10 lb of salt per gallon. Salt water with a concentration of 1/4 lb per gallon flows into the tank at the rate of 5 gallons per minute.

NOTE: Concentration = Salt / Volume

- (a) Express the total amount of salt after  $t$  minutes as an expression in  $t$ ; label this amount `Salt`. Use the `subs` command to evaluate `Salt` at  $t = 5, 12$  and 60 minutes.
- (b) Write the volume as an expression in  $t$  and evaluate it at  $t = 5, 12$  and 60 minutes.
- (c) Write the concentration as an expression in  $t$  and evaluate it at  $t = 5, 12$  and 60 minutes.
8. Enter  $f = \frac{x^5 + 1}{x^4 - 1}$  as an expression. Plot the expression  $f$  over the interval  $-2 \leq x \leq 0$ . Try using `subs` or `eval` to find  $f(-1)$ . Explain. Click in the plot at the point on the graph where  $x = -1$  to find  $\lim_{x \rightarrow -1} f$ . (When you click in the plot, the coordinates are shown at the top-left of the Maple window.) Then evaluate the limit. See `?Limit`.
9. Enter  $g(x) = \frac{x^2 - 9}{x^4 - 81}$ . Plot the graph of  $g$  over the interval  $1 \leq x \leq 5$ . Try to find  $g(3)$ . Explain. Then evaluate the limit  $\lim_{x \rightarrow 3} g(x)$ .
10. Consider the *expression* entered in Maple as `f:=x^2-1;`. Which of the following statements involve incorrect Maple syntax? How would you fix it?
- (a) `plot(f,x=-3..3);`      (b) `plot(f,-3..3);`  
 (c) `plot(f(x),x=-3..3);` (This is incorrect syntax, but works anyway.)  
 (d) `factor(f);`      (e) `subs(x=3,f);`      (f) `subs(x=3,f(x));`  
 (g) `eval(f(x), x=3);`      (h) `f(3);`      (i) `solve(f=3,x);`  
 (j) `solve(f(x)=3,x);`
11. Repeat Exercise 10 if  $f$  is now entered as the *function* `f:=x->x^2-1;`.
12. Enter each of the following as expressions:  $f = \frac{x-3}{3x-1}$  and  $g = \frac{x}{x+3}$ . Convert each expression to a function using `unapply`. Find the composition of the two functions by entering `f(g(x))` and by entering `(f@g)(x)` and simplify each. Here, the `@` symbol is used to indicate the composition, just as `o` is used in standard mathematical notation. Note that the output is an expression, since each function has been evaluated. Also try entering `h:=f@g;` which produces a function and then evaluating at `x`.

13. Graph  $g = \frac{\sqrt{4-x} - \sqrt{3+x}}{x^2 - 4}$  showing only  $-5 \leq x \leq 5$  and  $-10 \leq y \leq 10$ . Notice the vertical line segments that Maple puts in at the vertical asymptotes. Right click in the plot and select `STYLE > POINT` to see only the points Maple used and so verify that the lines are not really there. Now add the option `discont=true` to the plot, both without and with the option `style=point`. What happened?
14. Enter the expression  $g = \frac{\sqrt{4-x} - \sqrt{3+x}}{x^2 - 4}$ .
- Change the expression  $g$  into a function  $f$  by using `unapply`.
  - Find  $f(1), f(2), f(5)$ .
  - Find  $g(1), g(2), g(5)$  using `subs`.
  - Discuss what happens at  $x = 2$  and  $x = 5$ .
15. Form the expression  $f = \sqrt{x^2 \sin(x + \pi/4)}$ .
- Plot this expression for  $-2 \leq x \leq 4$  and  $-1 \leq y \leq 2$ . Click on the endpoints of the pieces to determine the domain of the expression. (When you click in the plot, the coordinates are shown in the top-left of the Maple window.) If the axes get in the way, right click in the plot and select `AXES > FRAMED` to move them.
  - Determine the domain of the expression to ten decimal places by using `fsolve` on  $f = 0$ , with ranges for  $x$  found from the plot.
16. Solve the following equations using the `solve` command. In each case, check your answers with a plot and by substituting the roots into the equation.
- $x^2 - 4x + 1 = 0$
  - $x^2 - 4.0x + 1.0 = 0$
  - $x^2 - x + 2 = 0$
  - $x^2 - x + 2.0 = 0$
  - $x^3 + 2x + 3 = 0$
  - $x^3 + 2.0x + 3.0 = 0$
17. Find all solutions to the following equations over the given interval.
- $\sin^2(x) = \cos(4x)$ , for  $0 \leq x \leq \pi$
  - $\cos^2(x) = \sin(4x)$ , for  $0 \leq x \leq \pi$
  - $8 \cos(x) = x$ , for  $-\infty < x < \infty$

HINT: Use the `plot` command to graph the two sides of each equation and click on each intersection. Then get a more accurate answer by using `fsolve`.

18. Solve the system of equations  $3x - 2y = 4$  and  $2x + 5y = 4$  for  $x$  and  $y$ .
19. Find the intersection point of the lines  $y = -\frac{3}{4}x + \frac{1}{4}$  and  $y = -\frac{1}{4}x + \frac{3}{4}$ .
20. Find the decimal approximations for all real roots of the equation

$$\frac{1}{x^4} + x^2 - 2x - 6 = 0$$

(Make sure you plot this one).

21. For each pair of expressions, plot them separately over the given intervals with different colors. Then combine them into a single plot using `display`.
- (a)  $f := -x^4$  for  $-1 \leq x \leq 1$  and  $g := 3x - 2$  for  $0 \leq x \leq 2$
- (b)  $f := x^3 - 2x^2$  for  $-2 \leq x \leq 3$ , and  $g := x^4 - 3x$  for  $-1 \leq x \leq 3$
- (c)  $f := \sin(x) + \cos(x)$  for  $-2\pi \leq x \leq \pi$  and  $g := \sin(x) - \cos(x)$  for  $-\pi \leq x \leq 2\pi$
22. Define the expression  $f = 25x + \frac{1}{x - 25}$  in Maple. Enter the following plot commands and discuss the behavior of each plot near  $x = 25$ .
- (a) `plot(f,x=0..100);`
- (b) `plot(f,x=0..100, numpoints=100);`
- (c) `plot(f,x=0..100, numpoints=10000);`
23. Use the `plot3d` command to view the graphs of the following expressions. Then click in the plot and drag the mouse.
- (a)  $f = x^2 + y^2$       (b)  $f = y^2 - x^2$       (c)  $f = x^4 - y^4$
- (d)  $f = (x^2 + y^2)(x - y)$       (e)  $f = \cos(x - y)$       (f)  $f = \sin(x) \sin(y)$
24. The point of this exercise is that plots can be deceiving. Enter the expression  $f := x^3 - x^2 - x + 1.001$ . Plot  $f$  over the interval  $-2 \leq x \leq 2$ . From this plot, guess how many real solutions there are to the equation  $f = 0$ . Now solve the equation  $f = 0$  with the `solve` command. How many solutions did `solve` return? Now replot  $f$  over the interval  $0.9 \leq x \leq 1.1$ . Does the graph of  $f$  cross the  $x$ -axis near the point  $x = 1$ ?
25. The point of this exercise is to show how piecewise defined functions can be entered into Maple using the `piecewise` command. The function

$$f(x) = \begin{cases} x^2 + 1 & x < 1 \\ 3 & x = 1 \\ x - 1 & \textit{otherwise} \end{cases}$$

can be entered into Maple as either the expression:

```
> f:=piecewise(x<1, x^2+1, x=1, 3, x-1);
```

or as the function

```
> g:=x->piecewise(x<1, x^2+1, x=1, 3, x-1);
```

The arguments of `piecewise` come in pairs consisting of a condition and a value, except that the last (optional) argument is the “otherwise” value. The expression `f` and the function `g` can now be evaluated. For example, the value of `f` at  $x = .5$  is

```
> eval(f,x=.5);
```

and the value of `g` at  $x = 1$  is

```
> g(1);
```

To plot `f` and `g` over the interval  $-1 \leq x \leq 3$  enter

```
> plot(f, x=-1..3, discont=true);
```

```
> plot(g, -1..3, discont=true);
```

NOTE: Maple is not reliable about plotting single points: There is a circle at  $(1, 3)$  in the graph of `f` but not in the graph of `g`.

Enter and plot the following functions with the `axes=framed` option.

$$p(x) = \begin{cases} \sin(x) & \text{if } x < 0 \\ \frac{1}{2} & \text{if } x = 0 \\ \cos(x) & \text{if } x > 0 \end{cases}$$

$$q(x) = \begin{cases} x^2 & \text{if } x \leq 0 \\ \sqrt{x} & \text{if } x > 2 \\ -2x & \text{otherwise} \end{cases}$$

NOTE:  $\leq$  is entered as `<=`.

26. Enter into Maple the equation of the circle with center at  $x = 2$  and  $y = 3$  with radius 2. Then graph the circle in two ways:
- `solve` the equation for  $y$  and label the solution. Plot the two solutions simultaneously using a single `plot` command. (Note that the circle looks more like an ellipse than a circle. What can you do to remove the distortion?)
  - Use `implicitplot` in the `plots` package to plot the equation directly.